California Wildlife Habitat Relationships Program
California Department of Fish and Game

HABITAT SUITABILITY MODELS FOR USE WITH ARC/INFO: MOUNTAIN BLUEBIRD



CWHR Technical Report No. 8
Sacramento, CA
June 1995

CWHR Technical Report No. 8
June 1995

HABITAT SUITABILITY MODELS FOR USE WITH ARC/INFO: MOUNTAIN BLUEBIRD

by

Irene C. Timossi
Ellen L. Woodard
Reginald H. Barrett
Department of Environmental Science, Policy, and Management
University of California
Berkeley, CA 94720
and the
Sierra Nevada Ecosystem Project

California Wildlife Habitat Relationships Program
Wildlife Management Division
California Department of Fish and Game
1416 Ninth Street
Sacramento, CA 95814

Suggested Citation: Timossi, I. C., E. L. Woodard, and R. H. Barrett. 1995. Habitat suitability models for use with ARC/INFO: Mountain bluebird. Calif. Dept. of Fish and Game, CWHR Program, Sacramento, CA. CWHR Tech. Report No. 8. 19 pp.

TABLE OF CONTENTS

PREFACE ii

ACKNOWLEDGMENTS v

HABITAT USE INFORMATION 1
General 1
Food 1
Water 1
Cover 1
Reproduction 2
Interspersion and Composition 2

HABITAT SUITABILITY INDEX (HSI) MODEL 2
Model Applicability 2
Geographic area. 2
Season. 2
Cover types. 3
Minimum habitat area. 3
Verification level. 3
Model Description 3
Overview. 3
Cover component. 4
Distance to water. 4
Species' distribution. 4
Spatial analysis 4
Definitions. 5
Application of the Model 6
Problems with the Approach 8
Cost. 8
Dispersal distance. 8
Day to day distance. 8

SOURCES OF OTHER MODELS 8

REFERENCES 9

APPENDIX 1: Mountain Bluebird Macro 10

MOUNTAIN BLUEBIRD (*Sialia currucoides*)

HABITAT USE INFORMATION

General

The mountain bluebird (*Sialia currucoides)* inhabits open coniferous forest, subalpine meadows, and pinyon-juniper (*Pinus-Juniperus*) woodlands from east-central Alaska, the western United States, and into northern Mexico (American Ornithologists' Union 1983). Throughout California, they are fairly common summer residents in sparse to open forests and woodlands and other open habitats from approximately 1,200 to 3,700 m (4,000 to 12,000 ft) (Zeiner et al. 1990). Most individuals winter below 1,500 m (5,000 ft) and are locally fairly common to abundant as winter residents in the Central Valley and surrounding foothills, agricultural areas of Owens and Antelope Valleys, and in Inyo, Kern, and Los Angeles counties. They are less numerous and their occurrence is more erratic in other interior lowlands of the state. During some years, mountain bluebirds may winter throughout the Mojave Desert, the southern coastal plains, and the Channel Islands (Grinnell and Miller 1944; Garrett and Dunn 1981).

Food

Though mountain bluebirds are primarily insectivorous, berries and other small fruits are eaten, particularly during winter (Martin et al. 1961; Power 1966). In Montana, three types of foraging behaviors were observed: perch-feeding; hovering; and flycatching (Power 1966). When perch-feeding, bluebirds flew from their perch to the ground where they would seize and consume a food item before returning to the perch. Hovering bluebirds flew up from their perch before rapidly descending to the ground to grab and consume their prey. Flycatching bluebirds sallied forth from perches to catch insects on the wing.

Water

Though no water requirements have been documented for this species, Miller and Stebbins (1964) and Power (1966) suggested that insects and green plant food provide adequate moisture.

Cover

The cover requirements of mountain bluebirds, which are satisfied by open forests and woodlands and other open habitats containing occasional trees and rocks, are similar to the food and reproductive requirements of the species (Grinnell and Miller 1944; Zeiner et al. 1990). In the Sierra Nevada, mountain bluebirds showed a preference for a burned stand of mixed coniferous forest over an unburned stand (Bock and Lynch 1970). Cover requirements are described under the Reproduction section below.

Reproduction

Mountain bluebirds breed in open coniferous and deciduous forests and woodlands, subalpine meadows, and other open habitats (American Ornithologists' Union 1983). Breeding pairs are most numerous where meadows, grasslands, or other open habitats abut woodland or rock formations affording suitable nesting sites. Eggs are laid from mid-April to mid-July depending on elevation, and two to three broods may be raised each season (Zeiner et al. 1990). Nests are concealed in old woodpecker holes, natural cavities in snags, or dead portions of trees. Less frequently, they are placed in rock crevices, holes in buildings, or in the nests of cliff swallows (*Hirundo pyrrhonota*) (Bent 1949).

Interspersion and Composition

Bock and Lynch (1970) found a nesting density of 15.2 pairs/40 ha (100 ac) in a burned stand of mixed-conifer forest in the Sierra Nevada. In Wyoming, nesting densities ranged from 15-18 individuals/40 ha (100 ac) (Finzel 1964) to 30 individuals/40 ha (100 ac) in aspen (*Populus tremuloides*) forests (Salt 1957). In Montana, territories ranged from 1.8-6.8 ha (4.5-16.7 ac) and averaged 4.3 ha (10.6 ac) (Power 1966). In Washington, a nesting female foraged over approximately 2.6 ha (6.5 ac) (Jewett et al. 1953). Territories are apparently centered on the nest and include suitable foraging perches and large open spaces (Power 1966).

HABITAT SUITABILITY INDEX (HSI) MODEL

Model Applicability

Geographic area.

The California Wildlife Habitat Relationships (CWHR) System (Airola 1988; Mayer and Laudenslayer 1988; Zeiner et al. 1990) contains habitat ratings for each habitat type predicted to be occupied by mountain bluebirds in California.

Season.

This model is designed to predict the suitability of habitats predicted to be used by mountain bluebirds, including breeding habitats.

Cover types.

This model can be used anywhere in California for which an ARC/INFO map of CWHR habitat types exists. The CWHR System contains suitability ratings for reproduction, cover, and feeding for all habitats mountain bluebirds are predicted to occupy. These ratings can be used in conjunction with the ARC/INFO habitat map to model wildlife habitat suitability.

Minimum habitat area.

Minimum habitat area is defined as the minimum amount of contiguous habitat required before a species will occupy an area. Specific information on minimum areas required for mountain bluebirds was not found in the literature. This model assumes two home ranges is the minimum area required to support a mountain bluebird population during the breeding season.

Verification level.

The spatial model presented here has not been verified in the field. The CWHR suitability values used are based on a combination of literature searches and expert opinion. We strongly encourage field testing of both the CWHR database and this spatial model.

Model Description

Overview.

This model uses CWHR habitat type as the main factor determining suitability of an area for this species.

A CWHR habitat type map must be constructed in ARC/INFO GRID format as a basis for the model. The GRID module of ARC/INFO was used for these models because of its superior functionality for spatial modeling. Only crude spatial modeling is possible in the vector portion of the ARC/INFO program, and much of the modeling done here would have been impossible without the abilities of the GRID module. In addition to more sophisticated modeling, the GRID module's execution speed is very rapid, allowing a complex model to run in less than 30 minutes.

The following sections document the logic and assumptions used to interpret habitat suitability.

Cover component.

A CWHR habitat map must be constructed. The mapped data (coverage) must be in ARC/INFO GRID format. A grid is a GIS coverage composed of a matrix of information. When the grid coverage is created, the size of the grid cell should be determined based on the resolution of the habitat data and the home range size of the species with the smallest home range in the study. You must be able to map the home range of the smallest species with reasonable accuracy. However, if the cell size becomes too small, data processing time can increase considerably. We recommend a grid cell size of 30 m (98 ft). Each grid cell can be assigned attributes. The initial map must have an attribute identifying the CWHR habitat type of each grid cell. A CWHR suitability value is assigned to each grid cell in the coverage based on its habitat type. Each CWHR habitat is rated as high, medium, low or of no value for each of three life requisites: reproduction; feeding; and cover. The geometric mean value of the three suitability values was used to determine the base value of each cell for this analysis.

Distance to water.

No water requirement was found for this species.

Species' distribution.

The study area must be manually compared to the range maps in the CWHR Species Notes (Zeiner et al. 1990) to ensure that it is within the species' range. All grid cells outside the species' range have a suitability of zero.

Spatial analysis.

Ideally, a spatial model of distribution should operate on coverages containing habitat element information of primary importance to a species. For example, in the case of woodpeckers, the size and density of snags as well as the vegetation type would be of great importance. For many small rodents, the amount and size of dead and down woody material would be important. Unfortunately, the large cost involved in collecting microhabitat (habitat element) information and keeping it current makes it likely that geographic information system (GIS) coverages showing such information will be unavailable for extensive areas into the foreseeable future.

The model described here makes use of readily available information such as CWHR habitat type, elevation, slope, aspect, roads, rivers, streams and lakes. The goal of the model is to eliminate areas that are unlikely to be utilized by the species and lessen the value of marginally suitable areas. It does not attempt to address all the microhabitat issues discussed above, nor does it account for other environmental factors such as toxins, competitors, or predators. If and when such information becomes available, this model could be modified to make use of it.

In conclusion, field surveys will likely discover that the species is not as widespread or abundant as predictions by this model suggest. The model predicts potentially available habitat. There are a variety of reasons why the habitat may not be utilized.

Definitions.

Home Range: the area regularly used for all life activities by an individual during the season(s) for which this model is applicable.

Dispersal Distance: the distance an individual will disperse to establish a new home range. In this model it is used to determine if Potential Colony Habitat will be utilized.

Day to Day Distance: the distance an individual is willing to travel on a daily or semi-daily basis to utilize a distant resource (Potential Day to Day Habitat). The distance used in the model is the home range radius. This is determined by calculating the radius of a circle with an area of one home range.

Core Habitat: a contiguous area of habitat of medium or high quality that has an area greater than two home ranges in size. This habitat is in continuous use by the species. The species is successful enough in this habitat to produce offspring that may disperse from this area to the Colony Habitat and Other Habitat.

Potential Colony Habitat: a contiguous area of habitat of medium or high quality that has an area between one and two home ranges in size. It is not necessarily used continuously by the species. The distance from a core area will affect how often Potential Colony Habitat is utilized.

Colony Habitat: Potential Colony Habitat that is within the dispersal distance of the species. These areas receive their full original value unless they are further than three home range radii from a core area. These distant areas receive a value of low since there is a low probability that they will be utilized regularly.

Potential Day to Day Habitat: an area of high or medium quality habitat less than one home range, or habitat of low quality of any size. This piece of habitat alone is too small or of inadequate quality to be Core Habitat.

Day to Day Habitat: Potential Day to Day Habitat that is close enough to a Core or Colony Habitat can be utilized by individuals moving out from those areas on a day to day basis. The grid cell must be within Day to Day Distance of Core or Colony Habitat.

Other Habitat: contiguous areas of low value habitat larger than two home ranges in size, including small areas of high and medium quality habitat that may be imbedded in them, are included as usable habitat by the species. Such areas may act as "sinks" because long-term reproduction may not match mortality.

The table below indicates the specific distances and areas assumed by this model.

Distance variables:	Meters	Feet
Dispersal Distance	2,279	7,478
Day to Day Distance/ Home Range radius	95	312

Area variables:	Hectares	M^2	Acres	Ft^2
Home Range	2.83	28,329	7	304,920
Core Habitat	5.66	56,658	14	609,840

Application of the Model

A copy of the ARC/INFO AML can be found in Appendix 1. The steps carried out by the macro are as follows:

1. **Determine Core Habitat**: this is done by first converting all medium quality habitat to high quality habitat and removing all low value habitat. Then contiguous areas of habitat are grouped into regions. The area of each of the regions is determined. Those large enough (two home ranges) are maintained in the Core Habitat coverage. If no Core Habitat is identified then the model will indicate no suitable habitat in the study area.

2. **Identify Potential Colony Habitat**: using the coverage from Step 1, determine which regions are one to two home ranges in size. These are Potential Colonies.

3. **Identify Potential Day Use Habitat**: using the coverage derived in Step 1, determine which areas qualify as Potential Day to Day Habitat.

4. **Calculate the Cost Grid**: Since it is presumed to be more difficult for animals to travel through unsuitable habitat than suitable habitat we use a cost grid to limit travel based on habitat suitability. The cost to travel is one for high or medium quality habitat. This means that to travel 1 m through this habitat costs 1 m of Dispersal Distance. The cost to travel through low quality habitat is two and unsuitable habitat costs four. This means that to travel 1 m through unsuitable habitat costs the species 4 m of Dispersal Distance.

5. **Calculate the Cost Distance Grid**: a cost distance grid containing the minimum cost to travel from each grid cell to the closest Core Habitat is then calculated using the Cost Grid (Step 4) and the Core Habitat (Step 1).

6. **Identify Colony Habitat**: based on the Cost Distance Grid (Step 5), only Potential Colony Habitat within the Dispersal Distance of the species to Core Habitat is retained. Colonies are close enough if **any** cell in the Colony is within the Dispersal Distance from Core Habitat. The suitability of any Colony located further than three home range radii from a Core Habitat is changed to low since it is unlikely it will be utilized regularly.

7. **Create the Core + Colony Grid**: combine the Core Habitat (Step 1) and the Colony Habitat (Step 6) and calculate the cost to travel from any cell to Core or Colony Habitat. This is used to determine which Potential Day to Day Habitat could be utilized.

8. **Identify Day to Day Habitat**: grid cells of Day to Day Habitat are only accessible to the species if they are within Day to Day Distance from the edge of the nearest Core or Colony Habitat. Add these areas to the Core + Colony Grid (Step 7).

9. **Add Other Habitat**: large areas (two home ranges in size) of low value habitat, possibly with small areas of high and medium habitat imbedded in them may be utilized, although marginally. Add these areas back into the Core + Colony + Day to Day Grid (Step 8), if any exist, to create the grid showing areas that will potentially be utilized by the species. Each grid cell contains a one if it is utilized and a zero if it is not.

10. **Restore Values**: all areas that have been retained as having positive habitat value receive

their original geometric mean value from the original geometric value grid (see *Cover component* section) with the exception of distant colonies. Distant colonies (colonies more than three home range radii distant) have their value reduced to low because of the low likelihood of utilization.

Problems with the Approach

Cost.

The cost to travel across low suitability and unsuitable habitat is not known. It is likely that it is quite different for different species. This model incorporates a reasonable guess for the cost of movement. A small bird will cross unsuitable habitat much more easily than a small mammal. To some extent differences in vagility between species is accounted for by different estimates of dispersal distances.

Dispersal distance.

The distance animals are willing to disperse from their nest or den site is not well understood. We have used distances from studies of the species or similar species when possible, otherwise first approximations are used. More research is urgently needed on wildlife dispersal.

Day to day distance.

The distance animals are willing to travel on a day to day basis to use distant resources has not been quantified for most species. This issue is less of a concern than dispersal distance since the possible distances are much more limited, especially with small mammals, reptiles, and amphibians. Home range size is assumed to be correlated with this coefficient.

SOURCES OF OTHER MODELS

No habitat models for mountain bluebird were found.

REFERENCES

Airola, D.A. 1988. Guide to the California Wildlife Habitat Relationship System. Calif. Dept. of Fish and Game, Sacramento, California. 74 pp.

American Ornithologists' Union. 1983. Check-list of North American birds. 6th ed. Allen Press, Lawrence, Kansas. 877 pp.

Bent, A.C. 1949. Life histories of North American thrushes, kinglets and their allies. U.S. Natl. Mus. Bull. 196. 454 pp.

Bock, C.E., and J.F. Lynch. 1970. Breeding bird populations of burned and unburned conifer forest in the Sierra Nevada. Condor 72:182-189.

Finzel, J.E. 1964. Avian populations of four herbaceous communities in southeastern Wyoming. Condor 66:496-510.

Garrett, K., and J. Dunn. 1981. Birds of southern California. Los Angeles Audubon Soc., Los Angeles, California. 408 pp.

Grinnell, J., and A.H. Miller. 1944. The distribution of the birds of California. Pac. Coast Avifauna No. 27. 608 pp.

Jewett, S.G., W.P. Taylor, W.T. Shaw, and J.W. Aldrich. 1953. Birds of Washington State. Univ. of Wash. Press, Seattle, Washington. 767 pp.

Martin, A.C., H.S. Zim, and A.L. Nelson. 1961. American wildlife and plants, a guide to wildlife food habits. Dover Publ., Inc., New York. 500 pp.

Mayer, K.E., and W.F. Laudenslayer, Jr., eds. 1988. A guide to wildlife habitats of California. Calif. Dept. of Fish and Game, Sacramento, California. 166 pp.

Miller, A.H., and R.C. Stebbins. 1964. The lives of desert animals in Joshua Tree National Monument. Univ. of Calif. Press, Berkeley, California. 452 pp.

Power, H.W., III. 1966. Biology of the mountain bluebird in Montana. Condor 68:351-371.

Salt, G.W. 1957. An analysis of avifaunas in the Teton Mountains and Jackson Hole, Wyoming. Condor 59:373-393.

Zeiner, D.C., W.F. Laudenslayer, Jr., K.E. Mayer, and M. White, eds. 1990. California's Wildlife. Vol. 2 Birds. Calif. Dept. of Fish and Game, Sacramento, California. 732 pp.

APPENDIX 1: Mountain Bluebird Macro

```
/*          MOUNTAIN BLUEBIRD

/* mtbmodel.aml - This macro creates an HSI coverage for the
/*               Mountain Bluebird.

/* Version:  Arc/Info 6.1 (Unix), GRID-based model.

/* Authors:  Irene Timossi, Sarah Miller, Wilde Legard,
/*          and Reginald H. Barrett
/*          Department of Forestry & Resource Management
/*          University of California, Berkeley

/* Revision: 2/10/95

/* ----------------------------------------------------------------

/* convert .ID to uppercase for info manipulations

&setvar .ID [translate %.ID%]

/* Start Grid

grid

/*

&type (1)  Initializing Constants...

/*   Homerange: the size of the species' homerange.

/*   DayPay: The amount the species is willing to pay traveling on
/*   a day-to-day basis.  Used to determine the area utilized on a
/*   day-to-day basis.

/*   DispersePay: Distance traveled when dispersing.  The amount
/*   the animal is willing to pay when dispersing from a core area.

/*   High:  The value in the WHR grid which indicates high quality habitat.

/*   Medium:  The value in the WHR grid which indicates medium quality habitat.

/*   Low:  The value in the WHR grid which indicates low quality habitat.

/*   None:  The value in the WHR grid which indicates habitat of no value.

/*   SpecCode: The WHR code for the species

/*   AcreCalc:  The number needed to convert square units
/*             (feet or meters) to acres.


&setvar SpecCode = B381

&if %.Measure% = Meters &then
  &do
    &setvar Homerange      = 28329
    &setvar DayPay         = 95
    &setvar DispersePay    = 2279
```

```
    &setvar AcreCalc       = 4047
  &end
&else
  &if %.Measure% = Feet &then
    &do
      &setvar Homerange      = 304920
      &setvar DayPay         = 312
      &setvar DispersePay    = 7478
      &setvar AcreCalc       = 43560
    &end
  &else
    &do
      &type Measurement type incorrect, check spelling.
      &type Only Meters and Feet are correct.
      &goto &BADEND
    &end

&setvar High          = 3
&setvar Medium        = 2
&setvar Low           = 1
&setvar None          = 0

/*   The following global variables are declared in the menu:

/*  .WHRgrid (WHR grid name): the name of the grid containing all
/*   the WHR information.

/*  .Bound (Boundary grid name): the grid containing only the
/*   boundary of the coverage. All cells inside the boundary
/*   have a value of 1. All cells outside the boundary must
/*   have a value < 1.

/*  .ID (Identifier): a 1 to 4 character code used to identify
/*   the files produced by this program. You may prefer
/*   to use an abbreviation of the species' common name
/*   (e.g. use `fis1` for fisher).

/*  .SizeOfCell (Cell size): the size (width) of the cells
/*   used in the coverage grids. All grids used in the
/*   analysis must have the same cell size.

/*  .Measure:  the units the coverage is measured in (feet or meters).


&type (2)  Creating working grid of geometric means...

/*    Create a Geometric Means grid (Geom) for the species by
/*    copying these values from the WHR grid.

Geom = %.WHRgrid%.%SpecCode%_G

/*

&type (3)  Changing %Medium% value cells to %High% value for Merge grid...

/*    Create a grid (Merge) merging Medium and High
/*    value cells from the Geometric mean grid (Geom),
/*    while leaving the value of other cells (Low and None) unchanged.
/*    Merge by changing the value of all medium cells to High.

Merge = con(Geom == %Medium%,%High%,Geom)
```

```
/*

&type (4)  Converting Merge grid zones into a Region grid...

/*    Convert the zones of the merge grid (Merge) into
/*    unique regions (Region). These will be used later
/*    to create core, colony, and day-to-day areas.

Region = regiongroup(Merge,#,EIGHT)

/*

&type (5)  Calculating the area of Region grid zones...

/*    Calculate the area of the zones (ZoneArea) on the region
/*    grid (Region).

ZoneArea = zonalarea(Region)

/*

&type (6)  Creating a Core Area grid...

/*    Extract areas from the zonal area grid (ZoneArea)
/*    suitable for core areas (Core). Core areas are defined
/*    as the Medium+High zones in the merge grid (Merge)
/*    with an area of at least two home ranges (%Homerange%).
/*    Set their value = 1.

if (Merge == %High% and ZoneArea >= %Homerange% * 2)
  Core = 1
endif

&if not [exists Core -vat] &then
  &goto END

/*

&type (7)  Creating a Colony grid...

/*    Extract areas from the zonal area grid (zoneArea)
/*    possibly suitable for colonization (ColTemp).
/*    Colony areas are defined as Low or Medium+High zones
/*    in the Merge grid (Merge) with an area of between one
/*    and two home ranges (%Homerange%). Set their value = 1.

/*    Then set all nodata values in the grid to zero (Colony).

docell
  if (Merge == %High%)
    if (ZoneArea > %Homerange% and ZoneArea < %Homerange% * 2)
      ColTemp = 1
    endif
  endif
end

Colony = con(isnull(ColTemp),0,ColTemp)

/*

&type (8)  Creating a Day-to-Day Use grid...
```

```
/*    Create a grid based on the values in the zonal
/*    area grid (ZoneArea) and merge grid (Merge)
/*    suitable for day-to-day use (DayToDay). Day-to-day use
/*    areas are defined as Low if the area is less than two
/*    homeranges in size or Medium+High zones in the
/*    merge grid (Merge) with an area of less than one home
/*    range (%Homerange%). Set their value = 1.

if ((Merge > %Low% and ZoneArea <= %Homerange%) or ~
    (Merge == %Low% and ZoneArea < %Homerange% * 2))
  DayToDay = 1
else
  DayToDay = 0
endif

/*

&type (9)  Creating a Cost Grid based on habitat value...

/*     Using the merge grid (Merge), create a cost grid (Cost)
/*     based on the habitat-value. Cost represents the relative
/*     resistance a species has to moving across different quality
/*     habitat:   Habitat-value   Cost
/*                None            4
/*                Low             2
/*                Medium+High     1

if (Merge == %None%)
   Cost = 4
else if (Merge == %Low%)
   Cost = 2
else if (merge == %High%)
  Cost = 1
endif

/*

&type (10) Calculating cost to travel from Core Areas...

/*     Calculate the cost to travel the distance (CostDist)
/*     from the nearest core area source (Core) using the cost
/*     grid (Cost).
/*

CostDist = CostDistance(Core,Cost)

/*

&type (11) Calculating which Colony areas are Cost Effective...

/*     If Colony Areas exist...
/*     Find the areas in the Colony grid (Colony) that could
/*     be colonized from the core areas:

/*     Assign costs to all cells in the Colony areas (Colony)
/*     from the Cost grid (CostDist). Zero surrounding NODATA areas.

/*     Make each colony a separate zone (ZoneReg) using
/*     the regiongroup command.
```

```
/*    Use zonalmin to find the minimum cost to arrive at each
/*    colony (ZoneMin).

/*    Set all NODATA cells to zero in ZoneMin to produce
/*    ColZer1.

/*    To find out which of the potential colonies can be utilized,
/*    determine which have a cost that is equal to or less than
/*    DispersePay. If the cost to get to a colony is less than
/*    or equal to DispersePay, keep it in grid Col.

/*    Fill the null value areas in Col with zeros to create ColZer2


&if not [exists ColTemp -vat] &then
  &goto SkipColony

ColDist = con(Colony > 0,CostDist,0)
ZoneReg = regiongroup(Colony,#,EIGHT)
ZoneMin = zonalmin(ZoneReg,ColDist)
ColZer1 = con(isnull(ZoneMin),0,ZoneMin)

if (ColZer1 <= %DispersePay% and ColZer1 > 0)
  Col = Colony
else
  Col = Core
endif

ColZer2 = con(isnull(Col),0,Col)

/*

&type (12)  Creating Core + Colony grid...

/*    If colonies exist....
/*    Create a grid (ColCore) that combines the core
/*    (Core) and colony (Colony) grids.
/*    This grid will be used to analyze day-to-day use.


if (Colony == 1)
  ColCore = 1
else
  ColCore = Core
endif

&label SkipColony


&type (13)  Calculate cost to travel from Core and Colony Areas...

/*   If colonies exist...
/*   Calculate the cost to travel the distance (CostDis2)
/*   from the nearest core or colony area source (ColCore).
/*   Otherwise just copy the CostDist grid to use for Day-to-Day
/*   analysis.

&if not [exists ColTemp -vat] &then
  CostDis2 = CostDist
&else CostDis2 = CostDistance(ColCore,Cost)
```

```
/*

&type (14) Calculating which Day-to-Day areas are Cost Effective...

/*     This step adds the utilized Day-to-Day cells to the
/*     Core + Colony Area grid (ColZer2) to produce the
/*     Day1 grid.

/*     Use the Core + Colony Cost grid (CostDis2)to find out
/*     what can actually be used day-to-day (any cell with
/*     a cost of DayPay or less).

/*     Retain any cell in the Day-to-Day grid (DayToDay) with
/*     a cost less than or equal to DayPay and greater than zero.

/*     If the Distance-Cost grid (CostDis2) = 0,
/*     it is part of the Core or Colony Area and
/*     should gets its value from Core + Colony Area
/*     grid (ColZer2).

&if [exists ColTemp -vat] &then
  &do
    if (CostDis2 <= %DayPay% and CostDis2 > 0)
      Day1 = DayToDay
    else
      Day1 = ColZer2
    endif
  &end
&else
  &do
    if (CostDis2 <= %DayPay% and CostDis2 > 0)
      Day1 = DayToDay
    else
      Day1 = Core
    endif
  &end

/*

&type (15)  Finding Other Areas That May Be Utilized....

/*     This step picks up any large low value areas and any small
/*     medium or high value polygons that are imbeded
/*     in them.

/*     First find any areas that are not currently in the included
/*     set (Day1Z) but are in the original geometric mean coverage (geom)
/*     set Other to 1 where Day1Z = 0.

/*     if Other is all nodata, create the All coverage from
/*     the Day1Z coverage.

/*     Split all Other areas into separate regions (OthReg)

/*     Calculate the area of the regions (OthArea).

/*     Keep any region in OthArea with an area > 2 homeranges (Util).

/*     Change any null values in Util to zeros (OthZero).

/*     Add these areas to the Day1 coverage to create All
```

```
Day1Z = con(isnull(Day1),0,Day1)

if ((Day1Z < 1) and (Geom > 0))
  Other = 1
endif

&if not [exists Other -vat] &then
  All = Day1Z
&else
  &do
    OthReg = regiongroup(Other,#,EIGHT)

    OthArea  = zonalarea(OthReg)

    if (OthArea >= %Homerange% * 2)
      Util = 1
    else
      Util = 0
    endif

    OthZero = con(isnull(Util),0,Util)

    if (OthZero == 1)
      All = OthZero
    else
      All = Day1Z
    endif
  &end

/*

&type (17) Creating a Value grid...

/*     For any cell in All that has a value of 1, store the suitability
/*     value from the Geometric mean grid (Geom) to the Value grid.

/*     Other cells inside the boundary (%.Bound%) get a value of 0.

/*

if (All == 1)
  Value = Geom
else if (%.Bound% == 1)
  Value = 0
endif

/*

&type (18)  Creating an HSI grid...

/*     if Colonies exist....
/*     For any cell that was part of a colony that is further than
/*     3 times the HR radius (DayPay) away from a core area, set the suitability
/*     to Low.  Distant colonies lose value because of their small size.
/*     This step produces grid Collow.

/*     Set all NODATA values in Collow to zero in ColZer3.

/*     Find any day-to-day use areas (DayToDay) that are being
/*     utilized (ColZer3).  If they are further than four homeranges
/*     from a core area (CostDist), they are utilized from a distant
```

```
/*     colony and their value will be decreased to Low in Day2.


/*     Then change nulls to zero in ValZero

/*     Keep all data within the boundary; call this final grid HSI.

&if [exists ColTemp -vat] &then
  &do
    if (ColZer1 >= %DayPay% * 3)
      Collow = %Low%
    else
      Collow = Value
    endif

    ColZer3 = con(isnull(Collow),0,Collow)

    if ((CostDist > %DayPay% * 4) and (ColZer3 > 0) and ~
        (DayToDay == 1))
      Day2 = 1
    else
      Day2 = ColZer3
    endif
  &end
&else
    Day2 = Value

valzero = con(isnull(Day2),0,Day2)

if (%.Bound% == 1)
  %.ID%hsi = valzero
endif

/*

&type (19)  Quiting from GRID and adding the acres field.....

/*     Quit from GRID (Q), then run additem to add an acre item to
/*     the HSI grid vat file (%ID%HSI.vat). Reindex on value when done.

Q
additem %.ID%hsi.vat %.ID%hsi.vat acres 10 10 i
indexitem %.ID%hsi.vat value

/*

&type (20)  Calculating acres.....

/*    Use INFO to calculate the acreage field: Multiply the number
/*    of cells by the cell size squared and divide by the number of
/*    square meters per acre (4047). Reindex on value when done.

&data arc info
arc
select %.ID%hsi.VAT
CALC ACRES = ( COUNT * %.SizeOfCell% * %.SizeOfCell% ) / %AcreCalc%
Q STOP

&END

indexitem %.ID%hsi.vat value
```

```
/*

&type  (21) Killing all intermediate coverages before ending macro...

/* &goto OKEND

grid

kill Geom
kill Merge
kill Region
kill ZoneArea
kill Core
kill ColTemp
kill Colony
kill DayToDay
kill Cost
kill CostDist
kill ColDist
kill ZoneReg
kill ZoneMin
kill ColZer1
kill Col
kill ColZer2
kill ColCore
kill CostDis2
kill Day1
kill Day1Z
kill Other
kill OthReg
kill OthArea
kill Util
kill OthZero
kill All
kill Value
kill Collow
kill ColZer3
kill Day2
kill valzero

q

&goto OKEND

&label END
&type **
&type **
&type NO CORE AREAS EXIST, EXITING MACRO
&type **
&type **

kill Core
kill Region
kill ZoneArea
kill Merge
kill Geom

quit

&label OKEND
&label BADEND
```

```
&type -–------------ All done! ----–-----------

&return
```